

Appendix A – Statistical code

Appendix A – Statistical code

Table of contents

<u>Dichotomous outcome</u>	2
<u>Continuous outcome (standardized mean difference)</u>	4
<u>Rate outcome</u>	6

Appendix A – Statistical code

Dichotomous outcome

```
model{

#Loop through studies
for(i in 1:NS) {
#Adjustment for multi-arm trials is zero for control arm
    w[i,1] <- 0
#Treatment effect is zero for control arm
    delta[i,1] <- 0

#Vague priors for trial baselines
    mu[i] ~ dnorm(0,0.0001)

#Binomial likelihood with logit link
for(k in 1:na[i]){
    x[i,k] ~ dbin(p[i,k],n[i,k])
    logit(p[i,k]) <- mu[i] + delta[i,k]
#Model fit
#Expected value of the numerators
    xhat[i,k] <- p[i,k] * n[i,k]
#Deviance contribution
    dev[i,k] <- 2 * (x[i,k] * (log(x[i,k])-log(xhat[i,k])))
    + (n[i,k]-x[i,k]) * (log(n[i,k]-x[i,k]) - log(n[i,k]-xhat[i,k])))
}

#Summed residual deviance contribution for each trial
resdev[i] <- sum(dev[i,1:na[i]])

#Trial-specific log(OR) distributions (random effect)
for(k in 2:na[i]){
    delta[i,k] ~ dnorm(md[i,k], taud[i,k])

#Mean of log(OR) distributions
    md[i,k] <- d[t[i,k]] - d[t[i,1]] + sw[i,k]

#Precision of log(OR) distributions
    taud[i,k] <- tau*2*(k-1)/k

#Adjustment for multi-arm RCTs
    w[i,k] <- (delta[i,k] - d[t[i,k]] + d[t[i,1]])}

#Cumulative adjustment for multi-arm RCTs
    sw[i,k] <- sum(w[i,1:k-1])/(k-1)
}

}
```

Appendix A – Statistical code

```
#Total residual deviance
totresdev <- sum(resdev[])

#Priors for treatment effects
d[1] <- 0 # treatment effect is zero for reference treatment
for(k in 2:NT){
    d[k] ~ dnorm(0,0.0001)
}

#Priors for random effects
#Main analysis
tau <- pow(sd,-2)
sd ~ dunif(0,2)
#Sensitivity analysis (additional uninformative prior)
#tau <- pow(sd,-2)
#sd ~ dnorm(0,1) T(0,)
#Sensitivity analysis (informative prior)
#log.variance ~ dnorm(-3.23, (1/(1.88*1.88)))
#variance <- exp(log.variance)
#sd <- sqrt(variance)
#tau <- 1/variance

#All pairwise treatment effects
#Probability that each treatment is superior to every other treatment
for(i in 1:NT) {
    for (j in 1:NT) { lor[i,j] <- d[i] - d[j]
                      log(OR[i,j]) <- lor[i,j]
                      better[i,j] <- step(OR[i,j]-1)
    }
}

#Rank the treatments
rk <- rank(d)

#Baseline treatment effect
for (k in 1:NTpbo){
    xPBO[k] ~ dbin(pPBO[k], nPBO[k])
    logit(pPBO[k]) <- muPBO[k]
    muPBO[k] ~ dnorm(mu0PBO, tauPBO)
}

tauPBO <- pow(sdPBO, -2)
sdPBO ~ dunif(0,2)
mu0PBO ~ dnorm(0,0.001)
```

Appendix A – Statistical code

Continuous outcome (standardized mean difference)

```
model{
#Loop through studies
for(i in 1:NS) {
#Adjustment for multi-arm trials is zero for control arm
w[i,1] <- 0
#Treatment effect is zero for control arm
delta[i,1] <- 0

#Vague priors for trial baselines
mu[i] ~ dnorm(0,0.0001)

#Normal likelihood with identity link
for(k in 1:na[i]){
    varia[i,k] <- pow(se[i,k],2)
    prec[i,k] <- 1/varia[i,k]
    x[i,k] ~ dnorm(theta[i,k],prec[i,k])
    theta[i,k] <- theta.std[i,k]*pooled.sd[i,k] #Standardization of the means
    theta.std[i,k] <- mu[i] + delta[i,k]

#Model fit
#Deviance contribution
    dev[i,k] <- (x[i,k]-theta[i,k])*(x[i,k]-theta[i,k])*prec[i,k]
}

#Summed residual deviance contribution for this trial
resdev[i] <- sum(dev[i,1:na[i]])

#Trial-specific distributions for the random effects
for(k in 2:na[i]){
    delta[i,k] ~ dnorm(md[i,k], taud[i,k])

#Mean of distributions
    md[i,k] <- d[t[i,k]] - d[t[i,1]] + sw[i,k]

#Precision of distributions
    taud[i,k] <- tau*2*(k-1)/k

#Adjustment for multi-arm trials
    w[i,k] <- (delta[i,k] - d[t[i,k]] + d[t[i,1]])}

#Cumulative adjustment for multi-arm RCTs
    sw[i,k] <- sum(w[i,1:k-1])/(k-1)
}
```

Appendix A – Statistical code

```
#Total residual deviance
totresdev <- sum(resdev[])

#Priors for treatment effects
d[1] <- 0 # treatment effect is zero for reference treatment
for(k in 2:NT){
    d[k] ~ dnorm(0,0.0001)
}

#Priors for random effects
tau <- pow(sd,-2)
sd ~ dunif(0,2)
#Sensitivity analysis (additional uninformative prior)
#tau <- pow(sd,-2)
#sd ~ dnorm(0,1) T(0,)
#Sensitivity analysis (informative prior)
#log.variance ~ dt(-2.7, (1/(2.46*2.46)),5)
#variance <- exp(log.variance)
#sd <- sqrt(variance)
#tau <- 1/variance

#All pairwise treatment effects
#Probability that each treatment is superior to every other treatment
#Convert the SMD into a mean difference, using the pooled SD of the reference scale
for(i in 1:NT) {
    for (j in 1:NT) {      SMD[i,j] <- (d[i] - d[j])
        better[i,j] <- step(SMD[i,j])
        diff[i,j] <- SMD[i,j]*mean.pooled.sd.ref
    }
}

#Rank the treatments
rk <- rank(d)

#Determine the baseline PBO risk
for(k in 1:NTpbo){
    xPBO[k] ~ dnorm(thetaPBO[k], precPBO[k])
    precPBO[k] <- 1/varPBO[k]
    varPBO[k] <- pow(sePBO[k],2)
    thetaPBO[k] ~ dnorm(mu0PBO, tauPBO)
}

tauPBO <- pow(sdPBO, -2)
sdPBO ~ dunif(0,2)
mu0PBO ~ dnorm(0,0.001)
```

Appendix A – Statistical code

Rate outcome

```
#Loop through studies
model{
  for(i in 1:NS) {
    #Adjustment for multi-arm trials is zero for control arm
    w[i,1] <- 0
    #Treatment effect is zero for control arm
    delta[i,1] <- 0

    #vague priors for trial baselines
    mu[i] ~ dnorm(0,0.0001)

    #Poisson likelihood with log link
    for(k in 1:na[i]){
      x[i,k] ~ dpois(lambda[i,k]*E[i,k])
      log(lambda[i,k]) <- mu[i] + delta[i,k]
    }
    #Model fit
    #Expected value of the numerators
    xhat[i,k] <- lambda[i,k]*E[i,k]
    #Deviance contribution
    x2[i,k] <- max(0.0000001,x[i,k])
    dev[i,k] <- -2*(xhat[i,k] - x2[i,k] + (x2[i,k]*log(xhat[i,k]/x2[i,k])))
  }

  #Summed residual deviance contribution for this trial
  resdev[i] <- sum(dev[i,1:na[i]])

  # Trial-specific distributions for the random effects
  for(k in 2:na[i]){
    delta[i,k] ~ dnorm(md[i,k],taud[i,k])

    # Mean of distributions
    md[i,k] <- d[t[i,k]] - d[t[i,1]] + sw[i,k]

    # Precision of distributions
    taud[i,k] <- tau*2*(k-1)/k

    #Adjustment multi-arm RCTs
    w[i,k] <- (delta[i,k] - d[t[i,k]] + d[t[i,1]])

    #Cumulative adjustment for multi-arm RCTs
    sw[i,k] <- sum(w[i,1:k-1])/(k-1)
  }
}

#Total residual deviance
```

Appendix A – Statistical code

```
totresdev <- sum(resdev[]])  
  
#Priors for treatment effects  
d[1] <- 0  
for(k in 2:NT){d[k] ~ dnorm(0,0.0001)}  
  
#Priors for random effects  
sd ~ dunif(0, 2)  
tau <- pow(sd,-2)  
#Sensitivity analysis (additional uninformative prior)  
#sd ~ dnorm(0,1) T(0,)  
#Sensitivity analysis (informative prior)  
#log.variance ~ dnorm(-3.23, (1/(1.88*1.88)))  
#variance <- exp(log.variance)  
#sd <- sqrt(variance)  
#tau <- 1/variance  
  
# All pairwise treatment effects  
# Probability that each treatment is superior to every other treatment  
for(i in 1:NT) {  
    for (j in 1:NT) { IRR[i,j] <- d[i] - d[j]  
        log(RR[i,j]) <- IRR[i,j]  
        better[i,j] <- step(RR[i,j]-1)  
    }  
}  
  
#Rank the treatments  
rk <- rank(d)  
  
#Determine the baseline PBO risk  
for (k in 1:NTpbo){  
    xPBO[k] ~ dpois(lambda.pbo[k]*E.PBO[k])  
    log(lambda.pbo[k]) <- muPBO[k]  
    muPBO[k] ~ dnorm(mu0PBO, tauPBO)  
}  
  
tauPBO <- pow(sdPBO, -2)  
sdPBO ~ dnorm(0,0.01)  
mu0PBO ~ dnorm(0,0.001)
```